MATH 308 Sheet 3

Maple can be used to solve differential equations and initial value problems. The key commands are dsolve, subs, rhs, simplify, evalf, and fsolve.

Example 1. $xy' + (x+1)y = 1$.
Assign the differential equation the name de for easy handling and remember to type the dependent variable y as $y(x)$.

```
> de:=x*diff(y(x),x)+(x+1)*y(x)=1;
```

Now tell Maple to solve the differential equation and assign the variable name sol to the solution. Notice the strange way Maple writes the constant of integration as $\_C1$.

```
> sol:=dsolve(de,y(x));
```

Notice the variable sol is actually an equation: To see this, type in

```
> sol;
```

To have Maple manipulate the solution, we need to make Maple ignore the "$y(x) =$" part. The way to do this is via the rhs command:

```
> rhs(sol);
```

For example, to have Maple check its work:

```
> subs(y(x)=rhs(sol),de);
```

This makes Maple plug its solution into the differential equation. Now expand and clean up (note the % sign tells Maple to plug in whatever is on the previous line for the % sign):

```
> simplify(%);
```

If all is well, this should give some kind of identity.

Example 2. Use the same differential equation, but now include the initial condition $y(2) = 1$. You already told Maple the differential equation, so you don't have to enter it again. Take note of how to enter the initial condition.

```
> sol:=dsolve({de,y(2)=1},y(x));
```

Sometimes it is necessary to "clean up" the solution. Here's one way:

```
> sol:=simplify(sol);
```

Check the solution as before and then check the initial condition as follows:

```
> subs(x=2,rhs(sol));
> simplify(%);
```

To plot the solution (notice the color is changed via the color=black option in contrast to linecolor=black for DEplot):

```
> plot(rhs(sol),x=0.5..5,color=black);
```

You can restrict the y range by including the y=a..b option, if necessary.

```
> plot(rhs(sol),x=0..5,y=0..50,color=black);
```

(What happens if you don't include the y=0..50 option in the preceeding line?)

a) Find the x value for which the solution has value 2. That is, solve the equation $y(x) = 2$ for x. Here's how to do it. First tell Maple the equation you want to solve:

```
> eq:=rhs(sol)=2;
```

Now tell Maple to find the x value:

```
> fsolve(eq,x);
```

b) Compute the value of $y(2/3)$. The first line tells Maple to plug in $x = 2/3$ to the solution and the next line tells it to compute the value.

```
> subs(x=2/3,rhs(sol));
> evalf(%);
```

c) Find the *positive* x value for which $y'(x) = -2$.

First have Maple compute the derivative and assign it to the variable der for easy handling:

```
> der:=diff(rhs(sol),x);
```

Enter the equation $y' = -2$:

```
> eq:=der=-2;
```

Now solve the equation:

```
> fsolve(eq,x);
```

Note this gives the *wrong* value: it's negative. So get the right root, restrict Maple's attention in the fsolve statement by using the option x=a..b. Use the graph of y to help determine a and b.

```
> fsolve(eq,x=0..infinity);
```

You could have also used x=0..4.

Example 3. Use dsolve to solve the initial value problem $x^2y' + xy = \sin x, \quad y(1) = 1$.
a) Plot the solution.
b) Compute $y(3)/4$.
c) Find $x > 0$ for which $y'(x) = -1$.
d) Find $x > 0$ for which $y'(x) = 1$.